

МИНИСТЕРСТВО НАУКИ И ВЫСШЕГО ОБРАЗОВАНИЯ
РОССИЙСКОЙ ФЕДЕРАЦИИ

Федеральное государственное автономное образовательное учреждение
высшего образования «Санкт-Петербургский политехнический университет
Петра Великого»

Институт компьютерных наук и кибербезопасности
Высшая школа технологий искусственного интеллекта
02.03.01 Математика и компьютерные науки

Курсовая работа
по дисциплине «Теория графов»
на тему «Модель оценки пригодности экзопланет
для колонизации по параметрам центрального
объекта системы»

Выполнил студент группы 5130201/40003 _____ Джабраилов А. В.

Проверил _____ Востров А. В.

«_____» _____ 2026 г.

Санкт-Петербург, 2026

Содержание

Введение	3
1 Предметная область	5
2 Математическое описание	7
2.1 Экспертные системы	7
2.2 Продукционная модель	8
2.3 Структура бинарного дерева	10
2.4 Механизм логического вывода	12
3 Особенности реализации	14
3.1 Среда CLIPS	14
3.2 Представление данных	14
3.3 Функция обработки пользовательского ввода	15
3.4 База знаний	16
3.5 Правила продукционной системы	16
3.6 Логика работы экспертной системы	18
Заключение	19
Приложение. Реализация экспертной системы в среде CLIPS	23

Введение

Экспертные системы — системы, предназначенные для неформализованных задач. Такие задачи трудно описать единым строгим алгоритмом, так как они могут содержать неполные, неточные или противоречивые данные, а также большое пространство возможных решений. Поэтому экспертная система опирается главным образом на эвристический поиск решения.

В рамках данной курсовой работы рассматривается задача оценки потенциальной обитаемости экзопланеты (далее — планеты). Эта задача также является неформализованной: на результат влияет совокупность факторов, например, тип центрального объекта, положение планеты относительно зоны обитаемости, наличие атмосферы, а также действие приливных сил и жёсткой радиации.

Целью работы является разработка экспертной системы, позволяющей по последовательности бинарных ответов пользователя классифицировать потенциальную обитаемость планеты. В качестве модели представления знаний выбрана продукционная модель, а структура принятия решений организована в виде бинарного дерева. Программная реализация выполнена в среде CLIPS.

Таким образом, в рамках данной работы необходимо:

1. Изучить предметную область.
 - Какие могут быть центральные объекты систем.
 - Каковы критерии, по которым можно оценить потенциальную обитаемость планеты, для каждого центрального объекта системы.
2. Построить бинарное дерево решений.
 - Дерево должно содержать минимум 4 яруса.
 - Дерево должно содержать минимум 30 узлов.
 - Факты, описанные в дереве, должны быть обоснованы авторитетными источниками.
3. Реализовать программу в среде CLIPS.

- Реализовать правила.
- Реализовать факты.
- Реализовать шаблоны и функции, необходимые для взаимодействия с пользователем и продвижению по дереву, реализованном в правилах и фактах.
- Реализовать устойчивость к некорректному вводу пользователя.

1. Предметная область

Разрабатываемая экспертная система предназначена для оценки потенциальной обитаемости планет в различных условиях. Под потенциальной обитаемостью в данной работе понимается возможность существования на планете условий, которые допускают наличие жидкой воды, устойчивой атмосферы и энергетического баланса, не приводящего к полному замерзанию поверхности или сильному радиационному фону.

Классический подход к определению обитаемости планет во многом связан с понятием зоны обитаемости [1]. Зона обитаемости представляет собой область вокруг центрального объекта, в которой планета при подходящих атмосферных условиях может поддерживать жидкую воду на поверхности.

В данной работе рассматриваются несколько типов центральных объектов и связанных с ними сценариев.

Во-первых, рассматриваются планеты около звёзд главной последовательности [1]. Наиболее благоприятными считаются стабильные звёзды с умеренной светимостью и относительно спокойной активностью. В разработанной системе отдельно учитываются звёзды с массой порядка $0.6\text{--}1.1M_{\odot}$, поскольку такие объекты могут достаточно долго поддерживать устойчивые условия излучения.

Во-вторых, отдельно рассматриваются планеты около М-карликов [1]. Эти звёзды имеют малую светимость, поэтому их зона обитаемости расположена близко к звезде. Такое положение повышает вероятность приливного захвата планеты. Кроме того, молодые М-карлики могут обладать высокой вспышечной активностью, что создаёт угрозу для атмосферы планеты.

В-третьих, в систему включены сценарии, связанные со сверхмассивными чёрными дырами [2]. В этом случае привычное понятие зоны обитаемости требует существенного расширения. Источником энергии может быть не только излучение звёздного типа, но и аккреционный диск или релятивистски усиленное фоновое излучение. При этом близость к чёрной дыре приводит к сильным приливным эффектам, замедлению времени и смещению внешнего излучения. Поэтому подобные сценарии в экспертной системе оцениваются осторожно и в большинстве случаев приводят к низкому потенциалу или непригодности.

В-четвёртых, рассматриваются планеты около нейтронных звёзд [3] и

малых чёрных дыр [2]. Для нейтронных звёзд существенными факторами являются рентгеновское излучение и гамма-излучение.

Также учитывается случай планеты около белого карлика после его охлаждения. У таких объектов зона обитаемости может быть очень узкой и расположенной близко к центральному телу, что может привести к приливному захвату. Поэтому даже при формальном выполнении температурных условий потенциал обитаемости в такой системе оценивается как низкий.

На основе этих соображений были выделены следующие основные группы критериев, используемых в экспертной системе:

1. тип центрального объекта;
2. масса и стабильность звезды, если центральный объект является звездой;
3. положение планеты относительно зоны обитаемости;
4. наличие атмосферы и магнитосферы;
5. степень активности центрального объекта;
6. вероятность приливного захвата или приливного разрушения;
7. влияние радиации и высокоэнергетических частиц;
8. наличие экстремальных релятивистских эффектов вблизи чёрной дыры.

Итогом работы экспертной системы является качественная экспертная оценка пригодности планеты для колонизации. Список возможных выводов экспертной системы:

1. высокий потенциал;
2. средний потенциал;
3. низкий потенциал;
4. непригодна.

2. Математическое описание

2.1. Экспертные системы

Экспертная система — это программная система, моделирующая рассуждения специалиста в некоторой узкой предметной области. В отличие от традиционной программы, которая обычно реализует заранее известный алгоритм, экспертная система предназначена для решения задач, где такой алгоритм трудно или невозможно задать полностью.

Сложные неформализованные задачи характеризуются следующими особенностями:

1. неполнотой, неточностью, неоднозначностью или противоречивостью исходных данных;
2. неполнотой и неоднозначностью знаний о предметной области;
3. большой размерностью пространства возможных решений;
4. изменчивостью данных и знаний непосредственно в процессе решения задачи.

Именно поэтому экспертные системы в значительной степени основаны на эвристическом поиске решения. Эвристика в данном случае понимается как правило или практический критерий, позволяющий сузить область поиска и приблизиться к разумному выводу без полного перебора всех возможных вариантов.

Экспертная система обычно включает следующие компоненты:

- интерфейс пользователя, через который осуществляется ввод данных и получение результата;
- базу знаний, содержащую правила, закономерности и сведения о предметной области;
- механизм логического вывода, сопоставляющий факты с правилами и формирующий новые факты или итоговое заключение;

Структура экспертной системы представлена на рис. 1.




Рис. 1: Структура экспертной системы.

Экспертная система работает в двух основных режимах. Первый режим — режим приобретения знаний. В нём база знаний формируется на основе выбранных источников и характеристик объектов. Второй режим — режим решения задачи, или режим консультации. В нём пользователь взаимодействует с системой, отвечает на вопросы и получает итоговый результат.

2.2. Продукционная модель

В качестве модели представления знаний используется продукционная модель. Продукционная система включает в себя три компонента:

1. набор правил, или продукций, представляющих базу знаний;
2. рабочую память, в которой хранятся исходные факты и факты, полученные в процессе вывода;
3. механизм логического вывода, применяющий правила к имеющимся фактам.

В основе продукционной модели находится продукция, то есть правило вида:

ЕСЛИ $\langle \text{условие} \rangle$ ТО $\langle \text{действие1} \rangle$ ИТАЧЕ $\langle \text{действие2} \rangle$.

Условие правила называется антецедентом, а действие — консеквен-
том. Условия могут объединяться с помощью логических связок AND и OR.
Правило срабатывает, если при сопоставлении фактов, находящихся в рабо-
чей памяти, с антецедентом правила обнаруживается совпадение. После это-
го результат выполнения правила заносится в рабочую память.

Структура правил продукционной модели изображена на рис. 2.




Рис. 2: Структура правил продукционной модели.

В данной экспертной системе продукционная модель используется для обхода бинарного дерева решений.

Правило для внутреннего узла можно записать следующим образом:

ЕСЛИ пользователь находится в узле Q_i1

И ответ пользователя равен *yes*,

ТО выполнить переход в узел $yes - next(Q_i2)$

ИНАЧЕ выполнить переход в узел $no - next(Q_i3)$

Для конечного узла (листа) правило имеет вид:

ЕСЛИ достигнут лист R_j ,

ТО вывести соответствующее заключение и завершить работу.

2.3. Структура бинарного дерева

Бинарное дерево решений представляет собой ориентированное дерево, в котором каждый узел имеет ровно два потомка: по веткам "да" и "нет".

Структура бинарного дерева решений экспертной системы представлена на рис. 3, рис. 4, рис. 5 и рис. 6.




Рис. 3: Бинарное дерево решений (часть 1).




Рис. 4: Бинарное дерево решений (часть 2).




Рис. 5: Бинарное дерево решений (часть 3).




Рис. 6: Бинарное дерево решений (часть 4).

Построенное дерево решений имеет следующие параметры:

Таблица 1: Параметры бинарного дерева решений

Параметр	Значение
Количество правил	17
Количество фактов	18
Общее количество узлов	35
Количество переходов между узлами	34
Количество уровней дерева	6
Минимальное количество ответов до результата	3
Максимальное количество ответов до результата	5
Количество различных итоговых заключений	4

2.4. Механизм логического вывода

Механизм логического вывода состоит в анализе имеющихся фактов и применении к ним правил из базы знаний для получения нового факта (узел)

или конечного результата (лист).

Алгоритм следующий:

1. в рабочую память помещается начальный факт, указывающий на корневое правило;
2. система находит описание текущего правила в базе знаний;
3. пользователю выводится текст вопроса;
4. пользователь вводит ответ;
5. система проверяет корректность ответа;
6. в зависимости от ответа выбирается следующая вершина дерева;
 - если следующая вершина является правилом, процесс повторяется;
 - если следующая вершина является фактом, система выводит заключение, а работа программы завершается.

3. Особенности реализации

3.1. Среда CLIPS

Для реализации экспертной системы была выбрана среда CLIPS. CLIPS является одним из языков построения экспертных систем, основанных на продукционной модели представления знаний.

В реализованной системе CLIPS используется для хранения базы знаний, организации рабочей памяти, обработки пользовательских ответов и выполнения переходов между узлами бинарного дерева, а также для взаимодействия с пользователем (получение ответов пользователя и возврат следующего узла дерева).

3.2. Представление данных

В программе используются шаблоны фактов, задаваемые с помощью конструкции `deftemplate`. Они описывают структуру данных, которые помещаются в рабочую память CLIPS.

Основными шаблонами являются:

- `answer` — хранит ответ пользователя на вопрос;
- `question` — описывает правило;
- `result` — описывает факт;
- `ask` — указывает, какой вопрос необходимо задать;
- `route` — задаёт переход к следующему узлу;
- `conclusion` — хранит итоговое заключение.

Например, шаблон вопроса имеет следующую структуру:

```
1 (deftemplate question
2   (slot id (type SYMBOL))
3   (slot text (type STRING))
4   (slot yes-next (type SYMBOL))
5   (slot no-next (type SYMBOL))
6 )
```

Поле `id` содержит идентификатор правила. Поле `text` хранит текст, выводимый пользователю. Поля `yes-next` и `no-next` задают идентификаторы узлов, к которым система должна перейти при ответах "да" и "нет" соответственно.

3.3. Функция обработки пользовательского ввода

Для ввода ответов используется функция `ask-yes-no`. Она выводит пользователю текст вопроса, считывает строку ответа, приводит её к нижнему регистру и проверяет, относится ли введённое значение к допустимым вариантам.

Функция принимает следующие варианты ответов в любом регистре:

- положительные ответы: да, д, yes, y;
- отрицательные ответы: нет, н, no, n.

Если пользователь вводит некорректное значение, система повторно запрашивает ответ.

Вход: строка (сообщение пользователя).

Выход: сообщение пользователя после парсинга или приглашение ввести ответ снова в корректной форме.

Код:

```
1 (deffunction ask-yes-no (?text)
2   (bind ?resp "")
3
4   (while TRUE do
5     (printout t ?text crlf "Введите ответ (да/нет): ")
6     (bind ?resp (lowercase (readline)))
7
8     (if (or (eq ?resp "да") (eq ?resp "д")
9             (eq ?resp "yes") (eq ?resp "y"))
10      then
11        (return yes))
12
13     (if (or (eq ?resp "нет") (eq ?resp "н")
14             (eq ?resp "no") (eq ?resp "n"))
15      then
```

```

16         (return no))
17
18     (printout t "Некорректный ввод." crlf crlf)
19 )
20 )

```

3.4. База знаний

База знаний экспертной системы задаётся с помощью конструкции `deffacts`. В ней перечислены все правила дерева и все возможные факты. Также в базе знаний задаётся начальное правило:

```

1 (ask (id Q1))

```

Этот факт указывает системе, что первым должен быть задан вопрос правила с идентификатором **Q1**, т.е. определяется корень бинарного дерева решений.

Пример описания вопроса в базе знаний:

```

1 (question
2   (id Q1)
3   (text "Центральный объект - звезда главной последовательности?")
4   (yes-next Q2)
5   (no-next Q3))

```

В данном примере, если пользователь отвечает на этот вопрос положительно, система переходит к правилу **Q2**, а если ответ отрицательный, выполняется переход к правилу **Q3**.

Пример описания конечного результата:

```

1 (result (id R2) (text "Высокий потенциал"))

```

3.5. Правила продукционной системы

В программе используется несколько универсальных правил, управляющих работой экспертной системы.

Правило `ask-question` отвечает за вывод вопроса пользователю, получение ответа и формирование факта перехода:

Вход: ID правила, вопрос которого нужно задать.

Выход: вопрос правила, который нужно задать.

Код:

```
1 (defrule ask-question
2   ?a <- (ask (id ?qid))
3   (question (id ?qid) (text ?text)
4             (yes-next ?yes-next)
5             (no-next ?no-next))
6   =>
7   (retract ?a)
8   (bind ?prompt (str-cat ?qid ": " ?text))
9   (bind ?resp (ask-yes-no ?prompt))
10  (assert (answer (question ?qid) (response ?resp)))
11
12  (if (eq ?resp yes)
13      then
14      (bind ?next ?yes-next)
15      else
16      (bind ?next ?no-next))
17
18  (assert (route (target ?next)))
19 )
```

После получения ответа система создаёт факт `route`, содержащий идентификатор следующего узла.

Итоговый вывод выполняется правилом `print-conclusion`:

Вход: вывод, который нужно вывести.

Выход: печать вывода на экран.

Код:

```
1 (defrule print-conclusion
2   ?c <- (conclusion (text ?text))
3   =>
4   (retract ?c)
5   (printout t crlf "Заключение: " ?text crlf)
6   )
```

3.6. Логика работы экспертной системы

Работа реализованной экспертной системы происходит последовательно. После запуска в рабочей памяти находится факт `ask` с идентификатором начального правила. Затем система извлекает соответствующее правило из базы знаний, выводит его пользователю и ожидает ответ.

В зависимости от ответа выбирается одна из двух ветвей дерева. Если выбранный узел является правилом, следующим шагом будет вывод нового вопроса пользователю и ожидание от него ответа. Если выбранный узел является фактом, система выводит итоговую оценку и программа завершает работу.

Алгоритм работы программы:

Вход: ответы пользователя на заданный ему вопрос.

Выход: оценка потенциальной обитаемости планеты.

1. запуск программы;
2. загрузка базы знаний;
3. активация начального правила Q1;
4. вывод вопроса пользователю;
5. получение и проверка ответа;
6. выбор следующего узла дерева;
7. если это правило, пункт 4; если это факт, пункт 8;
8. вывод конечного результата,
9. завершение программы

Заключение

В результате выполнения курсовой работы была разработана экспертная система оценки потенциальной обитаемости планеты. Система реализована в среде CLIPS и использует продукционную модель представления знаний. Логика принятия решений организована в виде бинарного дерева, узлы которого соответствуют правилам (т.е. вопросам), а листья — фактам (т.е. итоговой оценке).

Были выполнены следующие шаги:

1. Была изучена предметная область.
 - Было выяснено, какими могут быть центральные объекты систем.
 - Были изучены критерии, по которым можно оценить потенциальную обитаемость планеты, для каждого центрального объекта системы.
2. Было построено бинарное дерево решений.
 - Дерево содержит 6 ярусов.
 - Дерево содержит 35 узлов.
 - Факты, описанные в дереве, обоснованы авторитетными источниками; ссылки на источники даны в разделе Список литературы.
3. Была реализована программа в среде CLIPS.
 - Было реализовано 17 правил.
 - Было реализовано 18 фактов.
 - Были реализованы шаблоны и функции, необходимые для взаимодействия с пользователем и продвижению по дереву.
 - Была реализована устойчивость к некорректному вводу пользователя путём вывода сообщения с уведомлением о некорректном вводе и повторного запроса ввода в случае некорректного ввода.

Разработанное бинарное дерево включает 17 правил и 18 фактов. Итоговые результаты сведены к четырём видам: высокий потенциал, средний потенциал, низкий потенциал и непригодность.

Достоинства программы:

- база знаний системы является полной в рамках предметной области, рассматриваемой в ней;
- отделение базы знаний от правил логического вывода, что позволяет удобно добавлять новые знания без изменения правил вывода;
- возможность обработки некорректного пользовательского ввода;
- поддержку русских и английских вариантов ответов;
- сложность алгоритма — $\log(N)$.

Недостатки программы:

- отсутствие графического интерфейса;
- программа не допускает ответа "не знаю" или его аналогов, из чего следует, что пользователь должен знать ответ на каждый задаваемый ему вопрос, чтобы получить результат;
- реальные астрофизические условия сложнее, чем в данной модели: если экспертная система сделает вывод, что планета потенциально обитаема, на самом деле может оказаться, что она непригодна для жизни из-за других условий (например, характеристик самой планеты, а не объектов вокруг неё).
- при добавлении новых правил необходимо перестраивать всю часть дерева, идущую после добавленного правила.

Масштабируемость:

- добавление графического интерфейса;
- расширение базы знаний;
- упрощение процесса добавления новых знаний (например, через графический интерфейс);

- добавление поддержки ответа "не знаю": вопрос, на который пользователь не знает ответа, пропускается, и он отвечает на все вопросы, следующие из этого (например, планета может оказаться непригодна, основываясь на одном из ответов пользователя, независимо от ответов до этого вопроса);
- добавление поддержки обратного логического вывода.

Возможное развитие работы может включать расширение базы знаний, добавление режима неопределённого ответа, введение весовых коэффициентов для различных факторов, использование численных параметров планет и звёзд, а также создание графического интерфейса пользователя. Кроме того, систему можно дополнить модулем объяснения, который будет показывать пользователю путь по дереву и причины получения конкретного заключения.

Таким образом, разработанная программа демонстрирует применение методов теории графов, продукционных моделей и экспертных систем для решения задачи качественной оценки потенциальной обитаемости планет.

Список литературы

- [1] Michael Perryman. Exoplanet Handbook / Michael Perryman — 2-е изд. — Кембридж: 2011.— 3823 с.
- [2] Life on Miller's Planet: The Habitable Zone Around Supermassive Black holes [Электронный ресурс]. — URL: <https://arxiv.org/abs/1910.00940> (дата обращения: 21.04.2026).
- [3] Neutron Star Planets: Atmospheric processes and habitability [Электронный ресурс]. — URL: <https://arxiv.org/abs/1705.07688> (дата обращения: 21.04.2026).
- [4] Секция «Телематика» [Электронный ресурс]. — URL: <https://tema.spbstu.ru/tgraph/> (дата обращения: 11.05.2026).

Приложение. Реализация экспертной системы в среде CLIPS

```
1 (deftemplate answer
2   (slot question (type SYMBOL))
3   (slot response (type SYMBOL) (allowed-symbols yes no))
4 )
5
6 (deftemplate question
7   (slot id (type SYMBOL))
8   (slot text (type STRING))
9   (slot yes-next (type SYMBOL))
10  (slot no-next (type SYMBOL))
11 )
12
13 (deftemplate result
14   (slot id (type SYMBOL))
15   (slot text (type STRING))
16 )
17
18 (deftemplate ask
19   (slot id (type SYMBOL))
20 )
21
22 (deftemplate route
23   (slot target (type SYMBOL))
24 )
25
26 (deftemplate conclusion
27   (slot text (type STRING))
28 )
29
30 (deffunction ask-yes-no (?text)
31   (bind ?resp "")
32
33   (while TRUE do
34     (printout t ?text crlf "Введите ответ (да/нет): ")
35     (bind ?resp (lowercase (readline)))
36
37     (if (or (eq ?resp "да") (eq ?resp "д") (eq ?resp "yes") (eq ?resp
↪ "y"))
```

```

38     then
39     (return yes))
40
41     (if (or (eq ?resp "нет") (eq ?resp "н") (eq ?resp "no") (eq ?resp
42     ↪ "n"))
43     then
44     (return no))
45
46     (printout t "Некорректный ввод. Пожалуйста, введите \"да\" или
47     ↪ \"нет\".\" crlf crlf)
48 )
49 )
50 (deffacts knowledge-base
51   (question
52     (id Q1)
53     (text "Центральный объект – звезда главной последовательности?")
54     (yes-next Q2)
55     (no-next Q3))
56   (question
57     (id Q2)
58     (text "Масса звезды 0.6–1.1 М☉ (G / ранний-средний K-карлик)?")
59     (yes-next Q4)
60     (no-next Q5))
61   (question
62     (id Q3)
63     (text "Сверхмассивная чёрная дыра?")
64     (yes-next Q3a)
65     (no-next Q3b))
66   (question
67     (id Q4)
68     (text "Фотометрическая стабильность и умеренный ультрафиолетовый
69     ↪ фон?")
70     (yes-next Q6)
71     (no-next Q7))
72   (question
73     (id Q5)
74     (text "М-карлик массой < 0.6 М☉?")
75     (yes-next Q10)
76     (no-next Q11))
77   (question
78     (id Q6)

```


77 (text "Планета находится в консервативной зоне обитаемости?")
 78 (yes-next Q8)
 79 (no-next R1))
 80 (question
 81 (id Q7)
 82 (text "Повышенная хромосферная активность?")
 83 (yes-next Q9)
 84 (no-next R4))
 85 (question
 86 (id Q8)
 87 (text "Масса планеты 0.5–10 $M_{\oplus}$, есть атмосфера / магнитосфера /
 ↪ внутренняя активность?")
 88 (yes-next R2)
 89 (no-next R3))
 90 (question
 91 (id Q9)
 92 (text "У планеты есть сильная магнитосфера?")
 93 (yes-next R5)
 94 (no-next R6))
 95 (question
 96 (id Q10)
 97 (text "Приливный захват и высокая активность в первый 1 млрд лет?")
 98 (yes-next R7)
 99 (no-next R8))
 100 (question
 101 (id Q11)
 102 (text "Узкая зона обитаемости и быстрая эволюция звезды?")
 103 (yes-next R9)
 104 (no-next R10))
 105 (question
 106 (id Q3a)
 107 (text "Спин чёрной дыры экстремальный ($a/M \approx 1$)?")
 108 (yes-next Q3c)
 109 (no-next R11))
 110 (question
 111 (id Q3b)
 112 (text "Нейтронная звезда или малая чёрная дыра?")
 113 (yes-next Q3j)
 114 (no-next Q3k))
 115 (question
 116 (id Q3c)
 117 (text "Значительное замедление времени?")

```

118     (yes-next Q3d)
119     (no-next R12))
120 (question
121   (id Q3d)
122   (text "Блюшифт фонового излучения создаёт полезный поток энергии?")
123   (yes-next R13)
124   (no-next R14))
125 (question
126   (id Q3j)
127   (text "Орбита вне зоны приливного разрушения и жёсткой радиации?")
128   (yes-next R20)
129   (no-next R21))
130 (question
131   (id Q3k)
132   (text "Планета находится в узкой зоне обитаемости после охлаждения
133     ↪ белого карлика?")
134   (yes-next R22)
135   (no-next R23))
136
137 (result (id R1) (text "Непригодна"))
138 (result (id R2) (text "Высокий потенциал"))
139 (result (id R3) (text "Средний потенциал"))
140 (result (id R4) (text "Непригодна"))
141 (result (id R5) (text "Средний потенциал"))
142 (result (id R6) (text "Низкий потенциал"))
143 (result (id R7) (text "Низкий потенциал"))
144 (result (id R8) (text "Средний потенциал"))
145 (result (id R9) (text "Низкий потенциал"))
146 (result (id R10) (text "Средний потенциал"))
147 (result (id R11) (text "Низкий потенциал"))
148 (result (id R12) (text "Непригодна"))
149 (result (id R13) (text "Низкий потенциал"))
150 (result (id R14) (text "Непригодна"))
151 (result (id R20) (text "Низкий потенциал"))
152 (result (id R21) (text "Непригодна"))
153 (result (id R22) (text "Низкий потенциал"))
154 (result (id R23) (text "Непригодна"))
155
156 (ask (id Q1))
157 )
158 (defrule ask-question

```

```

159 ?a <- (ask (id ?qid))
160 (question (id ?qid) (text ?text) (yes-next ?yes-next) (no-next
    ↪ ?no-next))
161 =>
162 (retract ?a)
163
164 (if (eq ?qid Q1)
165     then
166     (printout t crlf
167      "Экспертная система оценки потенциальной обитаемости планеты."
168      ↪ crlf
169      "Отвечайте на вопросы словами \"да\" или \"нет\")." crlf))
170
171 (bind ?prompt (str-cat ?qid ": " ?text))
172 (bind ?resp (ask-yes-no ?prompt))
173
174 (assert (answer (question ?qid) (response ?resp)))
175
176 (if (eq ?resp yes)
177     then
178     (bind ?next ?yes-next)
179     else
180     (bind ?next ?no-next))
181
182 (assert (route (target ?next)))
183 )
184
185 (defrule route-to-question
186   ?r <- (route (target ?target))
187   (question (id ?target))
188   =>
189   (retract ?r)
190   (assert (ask (id ?target)))
191 )
192
193 (defrule route-to-result
194   ?r <- (route (target ?target))
195   (result (id ?target) (text ?text))
196   =>
197   (retract ?r)
198   (assert (conclusion (text ?text)))
199 )

```

```
199 (defrule print-conclusion
200   ?c <- (conclusion (text ?text))
201   =>
202   (retract ?c)
203   (printout t crlf "Заключение: " ?text crlf)
204 )
```